
Do programu „dopływa” strumień znaków — np. użytkownik wpisuje z klawiatury 
znak po znaku. 

Zadaniem programu jest wyznaczenie, ile wśród tych znaków jest dużych liter. 

Zakładamy, że wpisanie przez użytkownika znaku '\' kończy program.

Problem 0Problem 0

Problem 0 — analiza + pseudokodProblem 0 — analiza + pseudokod

Zdefiniuj zmienną całkowitą  — licznik dużych liter: LiczDL

Wyzeruj LiczDL

Powtarzaj
  Wczytaj znak
  Jeżeli znak jest dużą literą to
    Zwiększ LiczDL o 1
Aż zostanie naciśnięty znak '\'

Wyprowadź LiczDL



  

Copyright © Roman Simiński 4Strona :

Podstawy programowaniaPodstawy programowania Typy tablicowe — różne zastosowaniaTypy tablicowe — różne zastosowania

Problem 0 — przykładowe rozwiązanieProblem 0 — przykładowe rozwiązanie

Program LiczbaDuzychLiter;
Var
  LiczDL : Integer; (* Licznik duzych liter *)
  Znak   : Char;    (* Przechowuje wczytany znak *)
Begin
  WriteLn( 'Podawaj kolejne znaki (kazdy zatwierdz Enterem), ' );
  WriteLn( 'a ja policze ile wpisales duzych liter.' );
  WriteLn( 'Nacisnij \ by zakonczyc.' );

  (* Zerowanie licznika - bardzo wazna operacja *)
  LiczDL := 0;

  (* Iteracja wczytujaca znaki i zliczajaca duze litery *)
  Repeat
    Write( '>');
    ReadLn( Znak );
    If Znak In [ 'A' .. 'Z' ] Then
      LiczDL := LiczDL + 1;
  Until Znak = '\';

  (* Wyswietlenie licznosci duzych liter *)
  WriteLn( 'Liczba duzych liter: ', LiczDL );
  Write( 'Nacisnij Enter by zakonczyc program...' );
  ReadLn;
End.

Problem 0 — alternatywne wersje testu na dużą literęProblem 0 — alternatywne wersje testu na dużą literę

If Znak In [ 'A' .. 'Z' ] Then
  LiczDL := LiczDL + 1;

Wersja z wykorzystaniem zbioru znaków — testowanie przynależności znaku 
zapamiętanego w zmiennej Znak do zbioru dużych liter  [ 'A' .. 'Z' ]

If ( Znak >= 'A' ) And ( Znak <= 'Z' ] Then
  LiczDL := LiczDL + 1;

Testowanie, czy znak zapamiętany w zmiennej Znak, zawiera się w przedziale 
dużych liter typu porządkowego Char.

Problem 1Problem 1

Jak w Problem 0, tylko wyznaczamy liczbę wystąpień każdej z dużych liter.

Problem 1, rozwiązanie naiwne — analiza + pseudokod Problem 1, rozwiązanie naiwne — analiza + pseudokod 

Zdefiniuj zmienną całkowitą — licznik dla litery A: LiczA
Zdefiniuj zmienną całkowitą — licznik dla litery B: LiczB
...
Zdefiniuj zmienną całkowitą — licznik dla litery Z: LiczZ

Wyzeruj LiczA, LiczB, ... LiczZ

Powtarzaj
  Wczytaj znak
  Jeżeli znak jest dużą literą to
     Jeżeli znak to 'A' to 
        Zwiększ LiczA o 1
     Jeżeli znak to 'B' to 
        Zwiększ LiczB o 1
     ...
     Jeżeli znak to 'Z' to 
        Zwiększ LiczZ o 1
Aż zostanie naciśnięty znak '\'

Wyprowadź LiczA, LiczB, ... LiczZ

Problem 1, rozwiązanie naiwne — wątpliwościProblem 1, rozwiązanie naiwne — wątpliwości

Zdefiniuj zmienną całkowitą — licznik dla litery A: LiczA
Zdefiniuj zmienną całkowitą — licznik dla litery B: LiczB
...
Zdefiniuj zmienną całkowitą — licznik dla litery Z: LiczZ

Wyzeruj LiczA, LiczB, ... LiczZ

Potarzaj
  Wczytaj znak
  Jeżeli znak jest dużą literą to
     Jeżeli znak to 'A' to 
        Zwiększ LiczA o 1
     Jeżeli znak to 'B' to 
        Zwiększ LiczB o 1
     ...
     Jeżeli znak to 'Z' to 
        Zwiększ LiczZ o 1
Aż zostanie naciśnięty znak '\'

Wyprowadź LiczA, LiczB, ... LiczZ

Tyle zmiennych? Sporo 
pisania...

A jeżeli trzeba będzie 
jeszcze to samo zrobić 

dla małych liter i 
cyfr?!!  8-/

Problem 1, rozwiązanie naiwne — wątpliwości, cd. ...Problem 1, rozwiązanie naiwne — wątpliwości, cd. ...

Zdefiniuj zmienną całkowitą — licznik dla litery A: LiczA
Zdefiniuj zmienną całkowitą — licznik dla litery B: LiczB
...
Zdefiniuj zmienną całkowitą — licznik dla litery Z: LiczZ

Wyzeruj LiczA, LiczB, ... LiczZ

Potarzaj
  Wczytaj znak
  Jeżeli znak jest dużą literą to
     Jeżeli znak to 'A' to 
        Zwiększ LiczA o 1
     Jeżeli znak to 'B' to 
        Zwiększ LiczB o 1
     ...
     Jeżeli znak to 'Z' to 
        Zwiększ LiczZ o 1
Aż zostanie naciśnięty znak '\'

Wyprowadź LiczA, LiczB, ... LiczZ

Czy algorytm jest do ... 
znaczy, że kiepski? 

Nie, algorytm jest OK. 
A może trzeba to jakoś 

sprytnie 
zaprogramować... ?

Elementarz programisty: równanie Niklausa Wirth'aElementarz programisty: równanie Niklausa Wirth'a

Algorytm mam.

A o co chodzi z tymi 
strukturami danych... ?

algorytmy + struktury danych = programy



  

LiczDL

Problem 1, poszukiwanie rozwiązanie nienaiwnegoProblem 1, poszukiwanie rozwiązanie nienaiwnego

Rozwiązanie naiwne: definiujemy 
tyle liczników, ile dużych liter, każdy 
licznik jest osobną zmienną.

Rozwiązanie lepsze: definiujemy 
tablicę liczb całkowitych, tablica ma 
tyle elementów ile dużych liter.

Każdy element 

tablicy będzie licznikiem 

dla danej litery



  Świat użytkownika

Świat programisty

LiczDL

Problem 1, poszukiwanie rozwiązanie nienaiwnego, cd. ...Problem 1, poszukiwanie rozwiązanie nienaiwnego, cd. ...

A

B

Z

Wyświetl zawartość tablicy

Problem 1, jak zadeklarować tablicę liczników dla dużych liter?Problem 1, jak zadeklarować tablicę liczników dla dużych liter?

Type
  DuzeLitery = 'A' .. 'Z';
Var
  LiczDL : Array[ DuzeLitery ] Of Integer;

Var
  LiczDL : Array[ 'A' .. 'Z' ] Of Integer;

Deklaracja zmiennej tablicowej LiczDL:

Bardziej elegancka forma:

Jak odwołujemy się do elementów tablicy?

LiczDL[ 'A' ] := 0;
...
LiczDL[ 'A' ] := LiczDL[ 'A' ] + 1;
...
WriteLn( LiczDL[ 'A' ] );

Problem 1 — rozwiązanie sprytne Problem 1 — rozwiązanie sprytne 

Rozwiązanie bazuje na tym samy algorytmie co rozwiązanie naiwne, „sprytność” 
rozwiązania polega na mądrym wykorzystaniu możliwości typów tablicowych.

(* Zerowanie tablicy licznikow duzych liter *)
For Znak := 'A' To 'Z' Do
  LiczDL[ Znak ] := 0;

Zakładając, ze zmienna Znak jest typu Char, zerowanie wszystkich liczników:

LiczDL[ 'A' ] := 0;
LiczDL[ 'B' ] := 0;
LiczDL[ 'C' ] := 0;
.
.
.
LiczDL[ 'Z' ] := 0;

Co rozwija się w trakcie wykonania w ciąg instrukcji:

Problem 1 — rozwiązanie sprytne Problem 1 — rozwiązanie sprytne 

Write( '>');
ReadLn( Znak );
If Znak In [ 'A' .. 'Z' ] Then
  LiczDL[ Znak ] := LiczDL[ Znak ] + 1;

Wczytanie znaku do zmiennej Znak, sprawdzenie czy to duża litera, oraz 
zwiększenie licznika odpowiadającego wczytanej literze.

Repeat

  Write( '>');
  ReadLn( Znak );
  If Znak In [ 'A' .. 'Z' ] Then
    LiczDL[ Znak ] := LiczDL[ Znak ] + 1;

Until Znak = '\';

To samo powtarzane, aż do wczytania znacznika końca: 

Problem 1 — rozwiązanie sprytne Problem 1 — rozwiązanie sprytne 

WriteLn( 'Duze litery:' );

For Znak := 'A' To 'Z' Do
  Write( '|', Znak , ': ', LiczDL[ Znak ] : 4 );

Wyprowadzenie zawartości tablicy liczników dużych liter do strumienia 
wyjściowego programu: 

Problem 1 — przykładowy kod programu Problem 1 — przykładowy kod programu 

Program LiczenieDuzychLiter;
Type
  DuzeLitery = 'A' .. 'Z';
Var
  LiczDL : Array[ DuzeLitery ] Of Integer;
  Znak   : Char;
Begin
  (* Informacje o programie *)
  WriteLn( 'Podawaj kolejne znaki (kazdy zatwierdz Enterem), ' );
  WriteLn( 'a ja policze ile wpisales duzych liter.' );
  WriteLn( 'Nacisnij \ by zakonczyc.' );

  (* Zerowanie tablicy licznikow duzych liter *)
  For Znak := 'A' To 'Z' Do
    LiczDL[ Znak ] := 0;

  (* Iteracja wczytujaca i zliczajaca *)
  Repeat
    Write( '>');
    ReadLn( Znak );
    If Znak In [ 'A' .. 'Z' ] Then
      LiczDL[ Znak ] := LiczDL[ Znak ] + 1;
  Until Znak = '\';

(* Ciag dalszy >> *)

Problem 1 — rozwiązanie sprytne, przykładowy kod programu Problem 1 — rozwiązanie sprytne, przykładowy kod programu 

  (* Wyswietlenie liczby kazdej z duzych liter *)
  WriteLn;
  WriteLn( 'Duze litery:' );
  For Znak := 'A' To 'Z' Do
    Write( '|', Znak , ': ', LiczDL[ Znak ] : 4 );

  WriteLn;
  Write( 'Nacisnij Enter by zakonczyc program' );
  ReadLn;
End.

Dobór odpowiedniej struktury danych pozwolił zastosować naiwny i  — zdawało by się 
 — nieoptymalny algorytm.  

Często jest tak, że dobry pomysł na program nie tylko tkwi w algorytmie ale w dobrze 
zorganizowanych strukturach danych.

Problem 2Problem 2

Jak w Problem 1, tylko wyznaczamy liczbę wystąpień każdej z liter dużych, małych 
i cyfr.

Problem 2 — analogicznie jak dla dużych liter? Problem 2 — analogicznie jak dla dużych liter? 

Type
  DuzeLitery = 'A' .. 'Z';
  MaleLitery = 'a' .. 'z';
  Cyfry      = '0' .. '9';

Var
  LiczDL : Array[ DuzeLitery ] Of Integer;
  LiczML : Array[ MaleLitery ] Of Integer;
  LiczC  : Array[   Cyfry    ] Of Integer;

Var
  LiczDL : Array[ 'A' .. 'Z' ] Of Integer;
  LiczML : Array[ 'a' .. 'z' ] Of Integer;
  LiczC  : Array[ '0' .. '9' ] Of Integer;

Deklaracja zmiennej tablicowej LiczDL, LiczML, LiczC:

Bardziej elegancka forma:

Problem 2 — analogicznie jak dla dużych liter? Wątpliwości... Problem 2 — analogicznie jak dla dużych liter? Wątpliwości... 

Niby sprytne, ale będzie 
teraz trzeba męczyć się z 

trzema tablicami... :(

A jeżeli jeszcze dojdzie 
zapamiętywanie liczebności 
innych znaków, to robi się 

nieciekawie!

Problem 2 — a może wystarczy jedna tablica?Problem 2 — a może wystarczy jedna tablica?

Var
  Licznik : Array[ Char ] Of Integer;

Deklaracja tablicy liczników dla wszystkich znaków typy Char:

Przy tak zdefiniowanej tablicy, każdy znak będący elementem typu Char posiada swój 
własny licznik. Można zatem zliczać ile jest liter, cyfr, przecinków, kropek, itp., itd.

Dla naszych potrzeb wystarczy wykorzystać tylko wybrane elementy tablicy, reszta 
póki co się marnuje. Ale mimo tej redundancji, przetwarzanie jest proste, dotyczy 
jednej tablicy. 

0 A a1 . . .. . .
. . . . . . . . .. . .

. . . . . . . . .. . .
. . .

. . .
. . . . . .

B b9 Z z
Cyfry Duże litery Małe litery

0Kod: 48 49 57 65 66 90 97 98 122 255
Znak:

Uporządkowanie liter 
i cyfr w kodzie ASCII

spójne obszary kodowe

Problem 2 — wystarczy jedna tablicaProblem 2 — wystarczy jedna tablica

Program LiczenieLiterICyfr;
Var
  Licznik : Array[ Char ] Of Integer;
  Znak    : Char;
Begin
  (* Informacje o programie *)
  WriteLn( 'Podawaj kolejne znaki (kazdy zatwierdz Enterem), a ja policze' );
  WriteLn( 'ile wpisales kazdej z liter oraz kazdej cyfr.' );
  WriteLn( 'Nacisnij \ by zakonczyc.' );

  (* Zerowanie tablicy licznikow duzych liter, dziala dla ASCII *)
  For Znak := Chr( 0 ) To Chr( 255 ) Do
    Licznik[ Znak ] := 0;

  (* Iteracja wczytujaca i zliczajaca *)
  Repeat
    Write( '>');
    ReadLn( Znak );
    Licznik[ Znak ] := Licznik[ Znak ] + 1;
  Until Znak = '\';

(* Ciag dalszy >> *)



  

Copyright © Roman Simiński 24Strona :

Podstawy programowaniaPodstawy programowania Typy tablicowe — różne zastosowaniaTypy tablicowe — różne zastosowania

Problem 2 — wystarczy jedna tablicaProblem 2 — wystarczy jedna tablica

  (* Wyswietlenie licznika dla kazdej z duzych liter *)
  WriteLn;
  WriteLn( 'Duze litery:' );
  For Znak := 'A' To 'Z' Do
    Write( '|', Znak , ': ', Licznik[ Znak ] : 4 );

  (* Wyswietlenie licznika dla kazdej z malych liter *)
  WriteLn;
  WriteLn( 'Male litery:' );
  For Znak := 'a' To 'z' Do
    Write( '|', Znak , ': ', Licznik[ Znak ] : 4 );

  (* Wyswietlenie licznika dla kazdej z cyfr *)
  WriteLn;
  WriteLn( 'Cyfry:' );
  For Znak := '0' To '9' Do
    Write( '|', Znak , ': ', Licznik[ Znak ] : 4 );

  WriteLn;
  Write( 'Nacisnij Enter by zakonczyc program' );
  ReadLn;
End.

Kolejny program — wytypuj szóstkę do „dużego lotka” Kolejny program — wytypuj szóstkę do „dużego lotka” 

Zadaniem programu jest wylosowanie sześciu liczb, pozwalających wypełnić 
pojedynczy typowanie tzw. „dużego lotka”.

Losujemy liczby z przedziału 1 .. 49, w wylosowanej szóstce nie może być 
powtórzeń.

Wytypuj szóstkę do „dużego lotka” — analiza + pseudokod Wytypuj szóstkę do „dużego lotka” — analiza + pseudokod 

Zdefiniuj zmienną całkowitą  — licznik losowanych liczb: Wylosowanych
Zdefiniuj zmienną całkowitą  — pamięta aktualnie wylosowaną liczbę: Liczba

Zainicjuj generator liczb pseudolosowych
Wyzeruj zmienna Wylosowanych

Powtarzaj
   Wylosuj wartość z przedzialu 1 .. 49 i wstaw do zmiennej Liczba
   Wyprowadź zawartość zmiennej Liczba do strumienia wyjściowego
   Zwiększ zmienną Wylosowanych o 1
Aż zmienna Wylosowanych będzie miała wartość 6

To ma szansę zadziałać, ale nigdy nie wiadomo, czy losując kolejną liczbę, nie 
otrzymamy czasem którejś z już wcześniej wylosowanych liczb..., ale umówmy się, że 
ten problem rozwiążemy później.

Wytypuj szóstkę do „dużego lotka” — koślawa, pierwsza wersja Wytypuj szóstkę do „dużego lotka” — koślawa, pierwsza wersja 

Program LosujSzostke;
Const
  (* Losujemy liczby od 1 do wartosci stalej Zakres *)
  Zakres = 49;
Var
  Liczba        : Integer; (* Losowana liczba, kandydat do szostki *)
  Wylosowanych  : Integer; (* Licznik losowanych liczb  *)

Begin
  WriteLn( 'Losuje szostke liczb z 49-ciu.' );
  Randomize; (* Inicjalizacja generatora liczb pseudolosowych *)

  WriteLn( 'Wylosowana szostka: ' );
  Wylosowanych := 0; (* Zerowanie licznika wylosowanych liczb *) 
  Repeat
    Liczba := Random( Zakres ) + 1;  (* Losowanie liczby z zakresu *)
    Write( Liczba, ' ' ); (* Wyprowadzenie wylosowanej liczby *)
    Wylosowanych := Wylosowanych + 1;
  Until Wylosowanych = 6; (* Powtarzamy az wylosujemy szostke *)

  WriteLn;
  WriteLn( 'Nacisnij Enter by zakonczyc program...' );
  ReadLn;
End.

Wytypuj szóstkę do „dużego lotka” — jak wyeliminować duplikaty? Wytypuj szóstkę do „dużego lotka” — jak wyeliminować duplikaty? 

Zdefiniuj zmienną całkowitą  — licznik losowanych liczb: Wylosowanych
Zdefiniuj zmienną całkowitą  — pamięta aktualnie wylos. liczbę: Liczba

Zainicjuj generator liczb pseudolosowych
Wyzeruj zmienna Wylosowanych

Potarzaj
   Wylosuj wartość z przedzialu 1 .. 49 i wstaw do zmiennej Liczba
   Jeżeli wartość zmiennej Liczba nie była jeszcze wylosowana To
       Wyprowadź zawartość zmiennej Liczba
       Zwiększ zmienną Wylosowanych o 1
Aż zmienna Wylosowanych będzie miała wartość 6

W jaki sposób pamiętać, 
jakie liczby zostały 

wylosowane wcześniej?

Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb 

Każdy element rejestru odpowiada liczbie — kandydatce do szóstki. Jeżeli na 
danej pozycji w rejestrze ustawiona jest wartość Tak, to dana liczba została już 
wylosowana. 

Jeżeli na zadanej pozycji jest wartość Nie, taka liczba jeszcze nie była wylosowana. 

JuzWylosowane
Tak Nie Tak Nie Tak . . . Nie Nie Tak Tak Nie

1 2 3 4 5 45 46 47 48 49

Załóżmy, że wylosowano liczbę 45.

Na pozycji tej liczbie odpowiadającej jest Nie, 
zatem takiej liczby jeszcze nie wylosowano.

Liczba 45 może wejść do losowanej szóstki.  

Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb 

Każdy element rejestru odpowiada liczbie — kandydatce do szóstki. Jeżeli na 
danej pozycji w rejestrze ustawiona jest wartość Tak, to dana liczba została już 
wylosowana. 

Jeżeli na zadanej pozycji jest wartość Nie, taka liczba jeszcze nie była wylosowana. 

JuzWylosowane
Tak Nie Tak Nie Tak . . . Tak Nie Tak Tak Nie

1 2 3 4 5 45 46 47 48 49

Zaznaczamy, że liczba 45 została wylosowana 
wpisując do rejestru Tak, na pozycji 
odpowiadającej tej liczbie. 

Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb 

Każdy element rejestru odpowiada liczbie — kandydatce do szóstki. Jeżeli na 
danej pozycji w rejestrze ustawiona jest wartość Tak, to dana liczba została już 
wylosowana. 

Jeżeli na zadanej pozycji jest wartość Nie, taka liczba jeszcze nie była wylosowana. 

JuzWylosowane
Tak Nie Tak Nie Tak . . . Tak Nie Tak Tak Nie

1 2 3 4 5 45 46 47 48 49

Załóżmy, że wylosowano liczbę 5.

Na pozycji tej liczbie odpowiadającej jest Tak, 
zatem tę liczbę już wylosowano.

Liczba 5 zostaje odrzucona jako dublet.  

Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb Wytypuj szóstkę do „dużego lotka” — rejestr wylosowanych liczb 

Każdy element rejestru odpowiada liczbie — kandydatce do szóstki. Jeżeli na 
danej pozycji w rejestrze ustawiona jest wartość Tak, to dana liczba została już 
wylosowana. 

Jeżeli na zadanej pozycji jest wartość Nie, taka liczba jeszcze nie była wylosowana. 

JuzWylosowane
Tak Nie Tak Nie Tak . . . Tak Nie Tak Tak Nie

1 2 3 4 5 45 46 47 48 49

Z rejestru wylosowanych liczb można 
„wyczytać”, które z nich zostały wylosowane:

1  3  5  45  47  48  

Jak uzyskać taki rejestr w programie?Jak uzyskać taki rejestr w programie?

Taki rejestr to tablica 49 elementów. Każdy element takiej tablicy przyjmuje tylko 
jedną z dwóch możliwych wartości {Tak, Nie}, odpowiada to typowi Boolean 
w języku Pascal.

JuzWylosowane
Tak Nie Tak Nie Tak . . . Tak Nie Tak Tak Nie

1 2 3 4 5 45 46 47 48 49

Const
  Zakres = 49;

Var
  JuzWylosowane : Array[ 1 .. Zakres ] Of Boolean;

Jak sprawdzić, czy liczba już była (lub nie była) wylosowana?Jak sprawdzić, czy liczba już była (lub nie była) wylosowana?

(* Losowanie liczby z zakresu *)
Liczba := Random( Zakres ) + 1;

(* Czy juz te liczbe wylosowano? *)
If JuzWylosowane[ Liczba ] Then
  Begin
    (* Cos tam, gdy byla wylosowana *)
  End;

(* Losowanie liczby z zakresu *)
Liczba := Random( Zakres ) + 1;

(* Czy jeszcze tej liczby nie wylosowano? *)
If Not JuzWylosowane[ Liczba ] Then
  Begin
    (* Cos tam, gdy nie byla wylosowana *)
  End;



  

Copyright © Roman Simiński 35Strona :

Podstawy programowaniaPodstawy programowania Typy tablicowe — różne zastosowaniaTypy tablicowe — różne zastosowania

Jak oznaczyć, że liczba zostala wylosowana?Jak oznaczyć, że liczba zostala wylosowana?

(* Losowanie liczby z zakresu *)
Liczba := Random( Zakres ) + 1;

(* Czy jeszcze tej liczby nie wylosowano? *)
If Not JuzWylosowane[ Liczba ] Then
  Begin
    (* Nie bylo jeszcze takiej liczby, ustawiamy jej "flage" w tablicy *)
    JuzWylosowane[ Liczba ] := True;
  
    (* Cos tam dalej, gdy nie byla wylosowana *)
  End;

Jak wylosować całą szóstkę?Jak wylosować całą szóstkę?

(* Zerowanie licznika wylosowanych liczb *)
Wylosowanych := 0;

Repeat

  (* Losowanie liczby z zakresu *)
  Liczba := Random( Zakres ) + 1;

  (* Czy czasem juz tej liczby nie wylosowano ? *)
  If Not JuzWylosowane[ Liczba ] Then
    Begin
      (* Nie bylo jeszcze tej liczby, ustawiamy jej "flage" w tablicy *)  
      JuzWylosowane[ Liczba ] := True;

      (* Zwiekszamy licznik wylosowanych liczb *)
      Wylosowanych := Wylosowanych + 1;
    End;

    (* Powtarzamy az wylosujemy szostke *)
Until Wylosowanych = 6;

Jak wyprowadzić szóstkę?Jak wyprowadzić szóstkę?

(* Wyswietlenie wylosowanej szostki *)
WriteLn( 'Wylosowana szostka: ' );

For Liczba := 1 To Zakres Do  (* Przejrzyj rejestr liczb z zakresu *)
    If JuzWylosowane[ Liczba ] Then (* Jezeli liczba wylosowana *)
      Write( Liczba, ' ' ); (* Wyprowadz liczbe *)

Uwaga, żeby to wszystko zadziałało, trzeba wcześniej wyzerować rejestr 
wylosowanych liczb:

(* Jeszcze nie wylosowano zadnej liczby, czyszczenie tablicy *)
For Liczba := 1 To Zakres Do
  JuzWylosowane[ Liczba ] := False;

Wytypuj szóstkę do „dużego lotka” — pełna wersja programuWytypuj szóstkę do „dużego lotka” — pełna wersja programu

Program LosujSzostke;
Const
  (* Losujemy liczby od jeden do wartosci stalej Zakres *)
  Zakres = 49;
Var
  (* Tablica "flag" dla kazdej liczby z zakresu. Jezeli liczba zostala    *)
  (* wylosowana, to na jej pozycji bedzie wartosc True, False gdy liczba  *)
  (* nie zostala wylosowana                                               *)
  JuzWylosowane : Array[ 1 .. Zakres ] Of Boolean;
  
  (* Tutaj wylosowana liczba, kandydat do losowanej szostki *)
  Liczba        : Integer;
  
  (* Licznik losowanych liczb  *)
  Wylosowanych  : Integer;

Begin
  WriteLn( 'Losuje szostke liczb z 49-ciu.' );

  (* Jeszcze nie wylosowano zadnej liczby, czyszczenie tablicy *)
  For Liczba := 1 To Zakres Do
    JuzWylosowane[ Liczba ] := False;

  (* Inicjalizacja generatora liczb pseudolosowych *)
  Randomize; (* Ciag dalszy >> *)

Wytypuj szóstkę do „dużego lotka” — pełna wersja programu, cd.Wytypuj szóstkę do „dużego lotka” — pełna wersja programu, cd.

  (* Zerowanie licznika wylosowanych liczb *)
  Wylosowanych := 0;
  Repeat
    Liczba := Random( Zakres ) + 1; (* Losowanie liczby z zakresu *)
    If Not JuzWylosowane[ Liczba ] Then (* Czy wczesniej nie wylosowana? *)
      Begin
        (* Jeszcze niewylosowana, ustawiamy jej "flage" w tablicy *)
        JuzWylosowane[ Liczba ] := True;
        (* Zwiekszamy licznik wylosowanych liczb *)
        Wylosowanych := Wylosowanych + 1;
      End;
      (* Powtarzamy az wylosujemy szostke *)
  Until Wylosowanych = 6;

  (* Wyswietlenie wylosowanej szostki *)
  WriteLn( 'Wylosowana szostka: ' );
  For Liczba := 1 To Zakres Do  (* Sprawdz flagi liczb z zakresu *)
    If JuzWylosowane[ Liczba ] Then (* Jezeli liczba zostala wylosowana *)
      Write( Liczba, ' ' ); (* Wyprowadz liczbe *)

  WriteLn;
  WriteLn( 'Nacisnij Enter by zakonczyc program...' );
  ReadLn;
End.

Wytypuj szóstkę do „dużego lotka” — wersja alternatywnaWytypuj szóstkę do „dużego lotka” — wersja alternatywna

Zdefiniuj zmienną całkowitą — licznik losowanych liczb: Wylosowanych
Zdefiniuj zmienną całkowitą — pamięta aktualnie wylos. liczbę: Liczba
Zdefiniuj zmienną będącą zbiorem liczb — pamięta wylosowane liczby: Wylosowane  

Zainicjuj generator liczb pseudolosowych
Wylosowane := ∅

Potarzaj
   Wylosuj wartość z przedzialu 1 .. 49 i wstaw do zmiennej Liczba
   Jeżeli Liczba ∉ Wylosowane To
       Wylosowane := Wylosowane ∨ Liczba
       Zwiększ zmienną Wylosowanych o 1
Aż zmienna Wylosowanych = 6

Dla każdej liczby z zakresu Wykonaj
  Jeżeli Liczba ∈ Wylosowane To
     Wyprowadź Liczba



  

Copyright © Roman Simiński 41Strona :

Podstawy programowaniaPodstawy programowania Typy tablicowe — różne zastosowaniaTypy tablicowe — różne zastosowania

Zbiór jak struktura danychZbiór jak struktura danych

Typ zbiorowy należy do złożonych (strukturalnych) typów danych. 

Zbiór (zmienna teoriomnogościowa) składa się z elementów tego samego typu 
podstawowego, musi to być typ porządkowy. 

Zbiór może być pusty, lub zawierać elementy. 

Liczba elementów zbioru może się zmieniać.

Komputerowa implementacja zbiorów ma ograniczenia. Liczność zbioru nie 
może przekraczać pewnego limitu, zwykle jest to 256 elementów. 

Var
  Wylosowane : Set of Integer;

Zdefiniowanie takiej zmiennej się nie uda. Typ Integer ma zbyt wiele wartości:

Var
  DozwoloneZnaki : Set of Char;

Zdefiniowanie takiej zmiennej się powiedzie. Typ Char ma mniej wartości:

Definiowanie zbiorówDefiniowanie zbiorów

Var
  TrzyLitery : Set Of 'A' .. 'C';

Typ podstawowy musi być typem porządkowym.

Zmienna TrzyLitery może mieć wartości:

[], [ 'A' ], [ 'B' ], [ 'C' ], [ 'A', 'C' ], [ 'A', 'B' ], [ 'B', 'C' ],
[ 'A', 'B, 'C' ]

Wszystkie możliwe podzbiory zbioru podstawowego to zbiór potęgowy, 

Operacje na zbiorachOperacje na zbiorach

TrzyLitery := [];

Ustawienie zbioru pustego:

TrzyLitery := TrzyLitery + [ 'A' ];

Dopisanie elementu do zbioru:

Var
  Litera : 'A' .. 'C';
.
.
.
If Liter In TrzyLitery Then
  { cos tam };

Sprawdzenie, czy elemet należy do zbioru:

Na zbiorach działają operatory + (suma zbiorów),  ― (różnica zbiorów), 
* (przecięcie zbiorów), In (przynależność elementu do zbioru), = (równość 
zbiorów), <> (nierówność zbiorów), >= oraz <= (zawieranie zbiorów).

Definiowanie zbiorówDefiniowanie zbiorów

Var
  Wylosowane : Set Of 1 .. 49;

Aby zdefiniować zbiór o elementach typu liczba całkowita, należy wprowadzić typ 
okrojony: 

Zmienna Wylosowane może mieć przykładowe wartości:

  Wylosowane

[]

 [ 1, 2 ]

 [ 1, 2 ]
 [ 5, 21, 33 ]

 [ 49 ]

 [ 47, 48, 49 ]

Wytypuj szóstkę do „dużego lotka” — wykorzystanie zbiorówWytypuj szóstkę do „dużego lotka” — wykorzystanie zbiorów

Program LosujSzostke;
Const
  (* Losujemy liczby od jeden do wartosci stalej Zakres *)
  Zakres = 49;

Type
  (* Typ okrojony odpowiadajacy puli losowanych liczb   *)
  PulaLiczb = 1 .. Zakres;

Var
  (* Zbior losowanych liczb *)
  Wylosowane    : Set of PulaLiczb;

  (* Tutaj wylosowana liczba, kandydat do losowanej szostki *)
  Liczba        : PulaLiczb;

  (* Licznik losowanych liczb  *)
  Wylosowanych  : Integer;

(* Ciag dalszy >> *)

Wytypuj szóstkę do „dużego lotka” — wykorzystanie zbiorówWytypuj szóstkę do „dużego lotka” — wykorzystanie zbiorów

Begin
  WriteLn( 'Losuje szostke liczb z 49-ciu.' );

  (* Jeszcze nie wylosowano zadnej liczby, zbior wylosowanych ma byc pusty *)
  Wylosowane := [];

  (* Inicjalizacja generatora liczb pseudolosowych *)
  Randomize;
  (* Zerowanie licznika wylosowanych liczb *)
  Wylosowanych := 0;
  Repeat
    (* Losowanie liczby z zakresu *)
    Liczba := Random( Zakres ) + 1;
    (* Czy czasem juz tej liczby nie wylosowano ? *)
    If Not ( Liczba In Wylosowane ) Then
      Begin
        (* Nie bylo jeszcze takiej liczby, wstawiamy ja do zbioru *)
        Wylosowane := Wylosowane + [ Liczba ];
        (* Zwiekszamy licznik wylosowanych liczb *)
        Wylosowanych := Wylosowanych + 1;
      End;
      (* Powtarzamy az wylosoujemy szostke *)
  Until Wylosowanych = 6;

(* Ciag dalszy >> *)

  (* Wyswietlenie wylosowanej szostki *)
  WriteLn( 'Wylosowana szostka: ' );

  (* Sprawdz, ktore liczby z zakresu sa w zbiorze wylosowanych *)
  For Liczba := 1 To Zakres Do
    If Liczba In Wylosowane Then (* Jezeli liczba zostala wylosowana *)
      Write( Liczba, ' ' ); (* Wyprowadz liczbe *)

  WriteLn;
  WriteLn( 'Nacisnij Enter by zakonczyc program...' );
  ReadLn;
End.

